

Praktische Informatik 3: Funktionale Programmierung  
Vorlesung 5 vom 11.11.2014: Funktionen Höherer Ordnung I

Christoph Lüth

Universität Bremen

Wintersemester 2014/15

Rev. 2766

1 [33]

## Fahrplan

- ▶ Teil I: Funktionale Programmierung im Kleinen
  - ▶ Einführung
  - ▶ Funktionen und Datentypen
  - ▶ Rekursive Datentypen
  - ▶ Typvariablen und Polymorphie
  - ▶ Funktionen höherer Ordnung I
  - ▶ Funktionen höherer Ordnung II
  - ▶ Typinferenz
- ▶ Teil II: Funktionale Programmierung im Großen
- ▶ Teil III: Funktionale Programmierung im richtigen Leben

2 [33]

## Inhalt

- ▶ Funktionen höherer Ordnung:
  - ▶ Funktionen als gleichberechtigte Objekte
  - ▶ Funktionen als Argumente
- ▶ Spezielle Funktionen: map, filter, fold und Freunde

3 [33]

## Ähnliche Funktionen der letzten Vorlesung

- ▶ Pfade:

```
cat :: Path -> Path -> Path
cat Mt q      = q
cat (Cons i p) q = Cons i (cat p q)
```

```
rev :: Path -> Path
rev Mt      = ...
rev (Cons c l) = ...
```

Gelöst durch Polymorphie

```
cat :: MyString -> MyString -> MyString
cat Empty t      = t
cat (Cons c s) t = Cons c (cat s t)
```

```
rev :: MyString -> MyString
rev Empty      = Empty
rev (Cons c t) = cat (rev t) (Cons c Empty)
```

4 [33]

## Ähnliche Funktionen der letzten Vorlesung

```
kasse :: Einkaufswagen -> Int
kasse LeererWagen = 0
kasse (Einkauf a m e) = cent a m + kasse e
```

```
inventur :: Lager -> Int
inventur LeeresLager = 0
inventur (Lager a m l) = cent a m + inventur l
```

```
len :: MyString -> Int
len Empty      = 0
len (Cons c str) = 1 + len str
```


Gemeinsamkeiten:

- ▶ ein Fall pro Konstruktor
- ▶ linearer rekursiver Aufruf
- ▶ durch Polymorphie nicht gelöst (keine Instanz einer Definition)

5 [33]

## Muster der primitiven Rekursion

- ▶ Anwenden einer Funktion auf jedes Element der Liste
- ▶ möglicherweise Filtern bestimmter Elemente
- ▶ Kombination der Ergebnisse zu einem Gesamtergebnis E



6 [33]

## Ein einheitlicher Rahmen

- ▶ Zwei ähnliche Funktionen:

<pre>toL :: String -&gt; String toL []      = [] toL (c:cs) = toLower c : toL cs</pre>	<pre>toU :: String -&gt; String toU []      = [] toU (c:cs) = toUpper c : toL cs</pre>
----------------------------------------------------------------------------------------	----------------------------------------------------------------------------------------

- ▶ Warum nicht ...

```
map f []      = []
map f (c:cs) = f c : map f cs

toL cs = map toLower cs
toU cs = map toUpper cs
```

- ▶ Funktion f als Argument
- ▶ Was hätte map für einen Typ?

7 [33]

## Funktionen Höherer Ordnung

Slogan

"Functions are first-class citizens."

- ▶ Funktionen sind gleichberechtigt: Ausdrücke wie alle anderen
- ▶ Grundprinzip der funktionalen Programmierung
- ▶ Modellierung allgemeiner Berechnungsmuster
- ▶ Kontrollabstraktion

8 [33]

## Funktionen als Argumente: map

- map wendet Funktion auf alle Elemente an

- Signatur:

```
map :: (α → β) → [α] → [β]
```

- Definition wie oben

```
map f [] = []  
map f (x:xs) = f x : map f xs
```

- Auswertung:

```
toL "AB"  
→ map toLower ('A':'B':[]) → toLower 'A' : map toLower ('B':[])  
→ 'a':map toLower ('B':[]) → 'a':toLower 'B':map toLower []  
→ 'a':'b':map toLower [] → 'a':'b':[] ≡ "ab"
```

9 [33]

## Funktionen als Argumente: filter

- Elemente **filtern**: filter

- Signatur:

```
filter :: (α → Bool) → [α] → [α]
```

- Definition

```
filter p [] = []  
filter p (x:xs)  
| p x = x : filter p xs  
| otherwise = filter p xs
```

- Beispiel:

```
letters :: String → String  
letters = filter isAlpha
```

10 [33]

## Beispiel filter : Primzahlen

- Sieb des Eratosthenes

- Für jede gefundene Primzahl  $p$  alle Vielfachen heraus sieben
- Dazu: filter ( $\lambda n \rightarrow \text{mod } n \ p \neq 0$ ) ps
- Namenlose (anonyme) Funktion

```
sieve :: [Integer] → [Integer]  
sieve [] = []  
sieve (p:ps) = p : sieve (filter (\q → mod q p ≠ 0) ps)
```

- Alle Primzahlen:

```
primes :: [Integer]  
primes = sieve [2..]
```

- Die ersten  $n$  Primzahlen:

```
n_primes :: Int → [Integer]  
n_primes n = take n primes
```

11 [33]

## Funktionen als Argumente: Funktionskomposition

- Funktionskomposition (mathematisch)

```
(o) :: (β → γ) → (α → β) → α → γ  
(f o g) x = f (g x)
```

- Vordefiniert

- Lies:  $f$  nach  $g$

- Funktionskomposition **vorwärts**:

```
(>.>) :: (α → β) → (β → γ) → α → γ  
(f >.> g) x = g (f x)
```

- Nicht** vordefiniert!

12 [33]

## $\eta$ -Kontraktion

- Vertauschen der **Argumente** (vordefiniert):

```
flip :: (α → β → γ) → β → α → γ  
flip f b a = f a b
```

- Damit Funktionskomposition vorwärts:

```
(>.>) :: (α → β) → (β → γ) → α → γ  
(>.>) = flip (o)
```

- Da fehlt doch was?!** Nein:

```
(>.>) = flip (o) ≡ (>.>) f g a = flip (o) f g a
```

- Warum?

13 [33]

## $\eta$ -Äquivalenz und eta-Kontraktion

### $\eta$ -Äquivalenz

Sei  $f$  eine Funktion  $f : A \rightarrow B$ , dann gilt  $f = \lambda x. f x$

- Warum? **Extensionale** Gleichheit von Funktionen

- In Haskell:  **$\eta$ -Kontraktion**

- Bedingung: Ausdruck  $E :: \alpha \rightarrow \beta$ , Variable  $x :: \alpha$ ,  $E$  darf  $x$  nicht enthalten  
 $\lambda x \rightarrow E x \equiv E$

- Syntaktischer Spezialfall **Funktionsdefinition** (**punktfreie** Notation)

```
f x = E x ≡ f = E
```

14 [33]

## Partielle Applikation

- Funktionskonstruktor **rechtsassoziativ**:

```
a → b → c ≡ a → (b → c)
```

- Inbesondere**:  $(a \rightarrow b) \rightarrow c \neq a \rightarrow (b \rightarrow c)$

- Funktionsanwendung ist **linksassoziativ**:

```
f a b ≡ (f a) b
```

- Inbesondere**:  $f (a b) \neq (f a) b$

- Partielle** Anwendung von Funktionen:

- Für  $f :: a \rightarrow b \rightarrow c$ ,  $x :: a$  ist  $f x :: b \rightarrow c$  (**closure**)

- Beispiele:

- map toLower :: String → String
- (3 ==) :: Int → Bool
- concat o map (replicate 2) :: String → String

15 [33]

## Einfache Rekursion


- Einfache Rekursion**: gegeben durch

- eine Gleichung für die leere Liste
- eine Gleichung für die nicht-leere Liste (mit **einem** rekursiven Aufruf)

- Beispiel: kasse, inventur, sum, concat, length, (+), ...

- Auswertung:

```
sum [4,7,3] → 4 + 7 + 3 + 0  
concat [A, B, C] → A ++ B ++ C ++ []  
length [4, 5, 6] → 1 + 1 + 1 + 0
```



16 [33]

## Einfache Rekursion

► **Allgemeines Muster:**

```
f [] = A
f (x:xs) = x ⊗ f xs
```

► **Parameter der Definition:**

- Startwert (für die leere Liste)  $A :: \beta$
- Rekursionsfunktion  $\otimes :: \alpha \rightarrow \beta \rightarrow \beta$

► **Auswertung:**

```
f [x1,..., xn] = x1 ⊗ x2 ⊗ ... ⊗ xn ⊗ A
```

- **Terminiert** immer (wenn Liste endlich und  $\otimes, A$  terminieren)
- Entspricht einfacher Iteration (while-Schleife)

17 [33]

## Einfach Rekursion durch foldr

► **Einfache** Rekursion

- Basisfall: leere Liste
- Rekursionsfall: Kombination aus Listenkopf und Rekursionswert

► **Signatur**

```
foldr :: ( $\alpha \rightarrow \beta \rightarrow \beta$ )  $\rightarrow \beta \rightarrow [\alpha] \rightarrow \beta$ 
```

► **Definition**

```
foldr f e [] = e
foldr f e (x:xs) = f x (foldr f e xs)
```

18 [33]

## Beispiele: foldr

► **Summieren** von Listenelementen.

```
sum :: [Int]  $\rightarrow$  Int
sum xs = foldr (+) 0 xs
```

► **Flachklopfen** von Listen.

```
concat :: [[a]]  $\rightarrow$  [a]
concat xs = foldr (++) [] xs
```

► **Länge** einer Liste

```
length :: [a]  $\rightarrow$  Int
length xs = foldr ( $\lambda x n \rightarrow n + 1$ ) 0 xs
```

19 [33]

## Beispiele: foldr

► **Konjunktion** einer Liste

```
and :: [Bool]  $\rightarrow$  Bool
and xs = foldr (&&) True xs
```

► **Konjunktion** von Prädikaten

```
all :: ( $\alpha \rightarrow$  Bool)  $\rightarrow$  [ $\alpha$ ]  $\rightarrow$  Bool
all p = and  $\circ$  map p
```

20 [33]

## Der Shoppe, revisited.

► **Suche** nach einem Artikel alt:

```
suche :: Artikel  $\rightarrow$  Lager  $\rightarrow$  Maybe Menge
suche art (Lager (Posten lart m: l))
  | art == lart = Just m
  | otherwise = suche art (Lager l)
suche _ (Lager []) = Nothing
```

► **Suche** nach einem Artikel neu:

```
suche :: Artikel  $\rightarrow$  Lager  $\rightarrow$  Maybe Menge
suche a (Lager l) =
  listToMaybe (map ( $\lambda$ (Posten _ m)  $\rightarrow$  m)
               (filter ( $\lambda$ (Posten la _)  $\rightarrow$  la == a) l))
```

21 [33]

## Der Shoppe, revisited.

► **Kasse** alt:

```
kasse :: Einkaufswagen  $\rightarrow$  Int
kasse (Einkaufswagen []) = 0
kasse (Einkaufswagen (p: e)) = cent p + kasse (Einkaufswagen e)
```

► **Kasse** neu:

```
kasse' :: Einkaufswagen  $\rightarrow$  Int
kasse' (Einkaufswagen ps) = foldr ( $\lambda p r \rightarrow$  cent p + r) 0 ps

kasse :: Einkaufswagen  $\rightarrow$  Int
kasse (Einkaufswagen ps) = sum (map cent ps)
```

22 [33]

## Der Shoppe, revisited.

► **Kassenbon** formatieren neu:

```
kassenbon :: Einkaufswagen  $\rightarrow$  String
kassenbon ew@(Einkaufswagen as) =
  "Bob's_Aulde_Grocery_Shoppe\n\n" ++
  "Artikel_Menge_Preis\n" ++
  "-----\n" ++
  concatMap artikel as ++
  "-----\n" ++
  "Summe: " ++ formatR 31 (showEuro (kasse ew))
```

```
artikel :: Posten  $\rightarrow$  String
artikel p@(Posten a m) =
  formatL 20 (show a) ++
  formatR 7 (menge m) ++
  formatR 10 (showEuro (cent p)) ++ "\n"
```

23 [33]

## Noch ein Beispiel: rev

► **Listen umdrehen:**

```
rev :: [ $\alpha$ ]  $\rightarrow$  [ $\alpha$ ]
rev [] = []
rev (x:xs) = rev xs ++ [x]
```

► **Mit fold:**

```
rev' = foldr snoc []

snoc ::  $\alpha \rightarrow [\alpha] \rightarrow [\alpha]$ 
snoc x xs = xs ++ [x]
```

► **Unbefriedigend:** doppelte Rekursion  $O(n^2)$ !

24 [33]

## Einfache Rekursion durch foldl

- foldr faltet von rechts:

$$\text{foldr } \otimes [x_1, \dots, x_n] A = x_1 \otimes (x_2 \otimes (\dots (x_n \otimes A)))$$

- Warum nicht andersherum?

$$\text{foldl } \otimes [x_1, \dots, x_n] A = (((A \otimes x_1) \otimes x_2) \dots) \otimes x_n$$

- Definition von foldl:

```
foldl :: (α → β → α) → α → [β] → α
foldl f a [] = a
foldl f a (x:xs) = foldl f (f a x) xs
```

25 [33]

## Beispiel: rev revisited

- Listenumkehr ist falten von links:

```
rev' xs = foldl (flip (:)) [] xs
```

- Nur noch eine Rekursion  $O(n)$ !

26 [33]

## foldr vs. foldl

- $f = \text{foldr } \otimes A$  entspricht

```
f [] = A
f (x:xs) = x ⊗ f xs
```

- Kann nicht-strikt in xs sein, z.B. and, or
- Konsumiert nicht immer die ganze Liste
- Auch für nichtendliche Listen anwendbar

- $f = \text{foldl } \otimes A$  entspricht

```
f xs = g A xs
g a [] = a
g a (x:xs) = g (a ⊗ x) xs
```

- Endrekursiv (effizient) und strikt in xs
- Konsumiert immer die ganze Liste
- Divergiert immer für nichtendliche Listen

27 [33]

## foldl = foldr

### Definition (Monoid)

$(\otimes, A)$  ist ein **Monoid** wenn

```
A ⊗ x = x      (Neutrales Element links)
x ⊗ A = x      (Neutrales Element rechts)
(x ⊗ y) ⊗ z = x ⊗ (y ⊗ z)  (Assoziativität)
```

### Theorem

Wenn  $(\otimes, A)$  **Monoid**, dann für alle  $A, xs$

$$\text{foldl } \otimes A xs = \text{foldr } \otimes A xs$$

- Beispiele: length, concat, sum
- Gegenbeispiele: rev, all

28 [33]

## Übersicht: vordefinierte Funktionen auf Listen II

```
map    :: (α → β) → [α] → [β]    — Auf alle anwenden
filter :: (α → Bool) → [α] → [α]  — Elemente filtern
foldr  :: (α → β → β) → β → [α] → β — Falten v. rechts
foldl  :: (β → α → β) → β → [α] → β — Falten v. links
mapConcat :: (α → [β]) → [α] → [β] — map und concat
takeWhile :: (α → Bool) → [α] → [α] — längster Prefix mit p
dropWhile :: (α → Bool) → [α] → [α] — Rest von takeWhile
span    :: (α → Bool) → [α] → ([α], [α]) — take und drop
any     :: (α → Bool) → [α] → Bool — p gilt mind. einmal
all     :: (α → Bool) → [α] → Bool — p gilt für alle
elem    :: (Eq α) ⇒ α → [α] → Bool — Ist enthalten?
zipWith :: (α → β → γ) → [α] → [β] → [γ] — verallgemeinertes zip
```

29 [33]

## Funktionen Höherer Ordnung: Java

- Java**: keine direkte Syntax für Funktionen höherer Ordnung
- Folgendes ist **nicht** möglich:

```
interface Collection {
    Object fold(Object f(Object a, Collection c), Object a);
}
```

- Aber folgendes:

```
interface Foldable { Object f (Object a); }
interface Collection { Object fold(Foldable f, Object a); }
```

- Vergleiche Iterator aus Collections Framework (Java SE 6):

```
public interface Iterator<E> {
    boolean hasNext();
    E next();
}
```

- Seit Java SE 8 (März 2014): Anonyme Funktionen (Lambda-Ausdrücke)

30 [33]

## Funktionen Höherer Ordnung: C

- Implizit vorhanden: Funktionen = Zeiger auf Funktionen

```
extern list filter(int f(void *x), list l);
```

```
extern list map(void *(f)(void *x), list l);
```

- Keine direkte Syntax (e.g. namenlose Funktionen)
- Typsystem zu schwach (keine Polymorphie)
- Benutzung: qsort (C-Standard 7.20.5.2)

```
include <stdlib.h>
```

```
void qsort(void *base, size_t nmemb, size_t size,
           int (*compar)(const void *, const void *));
```

31 [33]

## Funktionen Höherer Ordnung: C

Implementierung von map:

```
list map(void *(f)(void *x), list l)
{
    list c;
    for (c = l; c != NULL; c = c->next) {
        c->elem = f(c->elem);
    }
    return l;
}
```

- Typsystem zu schwach:

```
{
    *(int *)x = *(int *)x*2;
    return x;
}
void prt(void *x)
```

```
printf("List: %u", mapM(prt, l)); printf("\n");
```

32 [33]

## Zusammenfassung

- ▶ Funktionen **höherer Ordnung**
  - ▶ Funktionen als **gleichberechtigte Objekte** und **Argumente**
  - ▶ Partielle Applikation,  $\eta$ -Kontraktion, namenlose Funktionen
  - ▶ Spezielle Funktionen höherer Ordnung: `map`, `filter`, `fold` und Freunde
- ▶ Formen der **Rekursion**:
  - ▶ Einfache Rekursion entspricht `foldr`